NAG Fortran Library Routine Document

F08PAF (DGEES)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08PAF (DGEES) computes the eigenvalues, the real Schur form T, and, optionally, the matrix of Schur vectors Z for an n by n real nonsymmetric matrix A.

2 Specification

```
SUBROUTINE F08PAF (JOBVS, SORT, SELECT, N, A, LDA, SDIM, WR, WI, VS,
1                  LDVS, WORK, LWORK, BWORK, INFO)
INTEGER            N, LDA, SDIM, LDVS, LWORK, INFO
double precision   A(LDA,*), WR(*), WI(*), VS(LDVS,*), WORK(*)
LOGICAL            SELECT, BWORK(*)
CHARACTER*1        JOBVS, SORT
EXTERNAL           SELECT
```

The routine may be called by its LAPACK name ***dgees***.

3 Description

The real Schur factorization of A is given by

$$A = ZTZ^T,$$

where Z is orthogonal, the matrix of Schur vectors, and T is quasi-upper triangular with 1 by 1 and 2 by 2 diagonal blocks.

A matrix is in real Schur form if it is upper quasi-triangular with 1 by 1 and 2 by 2 blocks. 2 by 2 blocks will be standardized in the form

$$\begin{bmatrix} a & b \\ c & a \end{bmatrix}$$

where $bc < 0$. The eigenvalues of such a block are $a \pm \sqrt{bc}$.

Optionally, F08PAF (DGEES) also orders the eigenvalues on the diagonal of the real Schur form so that selected eigenvalues are at the top left. The leading columns of Z then form an orthonormal basis for the invariant subspace corresponding to the selected eigenvalues.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBVS – CHARACTER*1 *Input*

On entry: if JOBVS = 'N', Schur vectors are not computed.

If JOBVS = 'V', Schur vectors are computed.

2: SORT – CHARACTER*1 *Input*

On entry: specifies whether or not to order the eigenvalues on the diagonal of the Schur form:

if SORT = 'N', eigenvalues are not ordered;
if SORT = 'S', eigenvalues are ordered (see SELECT).

3: SELECT – LOGICAL FUNCTION, supplied by the user. *External Procedure*

If SORT = 'S', SELECT is used to select eigenvalues to sort to the top left of the Schur form.

If SORT = 'N', SELECT is not referenced and F08PAF (DGEES) may be called with the dummy function F08PAZ.

Its specification is:

```
LOGICAL FUNCTION SELECT (WR, WI)
double precision            WR, WI
```

1: WR – ***double precision*** *Input*
2: WI – ***double precision*** *Input*

On entry: an eigenvalue $\mathrm{WR}(j) + \sqrt{-1} \times \mathrm{WI}(j)$ is selected if $\mathrm{SELECT}(\mathrm{WR}(j), \mathrm{WI}(j))$ is .TRUE.. If either one of a complex conjugate pair of eigenvalues is selected, then both complex eigenvalues are selected. Note that a selected complex eigenvalue may no longer satisfy $\mathrm{SELECT}(\mathrm{WR}(j), \mathrm{WI}(j)) =$ .TRUE. after ordering, since ordering may change the value of complex eigenvalues (especially if the eigenvalue is ill-conditioned); in this case INFO is set to $\mathrm{N} + 2$ (see INFO below).

SELECT must be declared as EXTERNAL in the (sub)program from which F08PAF (DGEES) is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

On exit: has been overwritten by its real Schur form T.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08PAF (DGEES) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

7: SDIM – INTEGER *Output*

On exit: if SORT = 'N', SDIM = 0.

If SORT = 'S', SDIM = number of eigenvalues (after sorting) for which SELECT is true. (Complex conjugate pairs for which SELECT is true for either eigenvalue count as 2.)

8: WR($*$) – ***double precision*** array *Output*

Note: the dimension of the array WR must be at least $\max(1, \mathrm{N})$.

On exit: see the description of WI below.

9: WI($*$) – ***double precision*** array *Output*

Note: the dimension of the array WI must be at least $\max(1, \mathrm{N})$.

On exit: WR and WI contain the real and imaginary parts, respectively, of the computed eigenvalues in the same order that they appear on the diagonal of the output Schur form T. Complex conjugate pairs of eigenvalues will appear consecutively with the eigenvalue having the positive imaginary part first.

10: VS(LDVS,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array VS must be at least $\max(1, \mathrm{N})$ if JOBVS = 'V' and at least 1 otherwise.

On exit: if JOBVS = 'V', VS contains the orthogonal matrix Z of Schur vectors.

If JOBVS = 'N', VS is not referenced.

11: LDVS – INTEGER *Input*

On entry: the first dimension of the array VS as declared in the (sub)program from which F08PAF (DGEES) is called.

Constraints:

if JOBVS = 'V', LDVS $\geq \max(1, \mathrm{N})$;
LDVS ≥ 1 otherwise.

12: WORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) contains the optimal LWORK.

13: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08PAF (DGEES) is called.

For good performance, LWORK must generally be larger than the minimum, say $3 \times \mathrm{N} + nb \times \mathrm{N}$, where nb is the optimal block size for F08NEF (DGEHRD)

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraint: LWORK $\geq \max(1, 3 \times \mathrm{N})$.

14: BWORK($*$) – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least 1 if SORT = 'N' and at least $\max(1, \mathrm{N})$ otherwise.

If SORT = 'N', BWORK is not referenced.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

If INFO $= i$ and $i \leq$ N, the QR algorithm failed to compute all the eigenvalues

INFO $=$ N $+ 1$

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the problem is very ill-conditioned).

INFO $=$ N $+ 2$

After reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues in the Schur form no longer satisfy SELECT $=$ .TRUE.. This could also be caused by underflow due to scaling.

7 Accuracy

The computed Schur factorization satisfies

$$A + E = ZTZ^T,$$

where

$$\|E\|_2 = \mathrm{O}(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.8 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3.

The complex analogue of this routine is F08PNF (ZGEES).

9 Example

To find the Schur factorization of the matrix

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix},$$

such that the real eigenvalues of A are the top left diagonal elements of the Schur form, T.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08PAF Example Program Text
*     Mark 21.  NAG Copyright 2004
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDVS, LWORK
      PARAMETER        (LDA=NMAX,LDVS=NMAX,LWORK=(2+NB)*NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N, SDIM
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), VS(LDVS,NMAX), WI(NMAX),
```

```
     +                 WORK(LWORK), WR(NMAX)
      LOGICAL          BWORK(NMAX)
*     .. External Functions ..
      LOGICAL          SELECT
      EXTERNAL         SELECT
*     .. External Subroutines ..
      EXTERNAL         DGEES, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08PAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the matrix A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*        Find the Schur factorization
*
         CALL DGEES('Vectors (Schur)','Sort',SELECT,N,A,LDA,SDIM,WR,WI,
     +             VS,LDVS,WORK,LWORK,BWORK,INFO)
         LWKOPT = WORK(1)
*
         IF (INFO.EQ.0 .OR. INFO.EQ.(N+2)) THEN
*
*           Print solution
*
            WRITE (NOUT,99999)
     +        'Number of eigenvalues for which SELECT is true = ', SDIM
            WRITE (NOUT,*)
            IF (INFO.EQ.(N+2)) THEN
               WRITE (NOUT,99998) '***Note that rounding errors mean ',
     +           'that leading eigenvalues in the Schur form',
     +           'no longer satisfy SELECT = .TRUE.'
               WRITE (NOUT,*)
            END IF
*
*           Print out factors of the Schur factorization
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,N,A,LDA,'Schur matrix T',IFAIL)
*
            WRITE (NOUT,*)
            CALL X04CAF('General',' ',N,N,VS,LDVS,
     +                  'Matrix of Schur vectors Z',IFAIL)
         ELSE
            WRITE (NOUT,99997) 'Failure in DGEES.  INFO = ', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99996) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I4)
99998 FORMAT (1X,2A,/1X,A)
99997 FORMAT (1X,A,I4)
99996 FORMAT (1X,A,I5,/1X,A,I5)
      END

      LOGICAL FUNCTION SELECT(AR,AI)
```

```
*     .. Scalar Arguments ..
*
*     Logical function SELECT for use with DGEES (F08PAF)
*
*     Returns the value .TRUE. if the imaginary part of the eigenvalue
*     (AR + AI*i) is zero, i.e. the eigenvalue is real
*
      DOUBLE PRECISION        AI, AR
*     .. Local Scalars ..
      LOGICAL                 D
*     .. Executable Statements ..
      IF (AI.EQ.0.0D0) THEN
         D = .TRUE.
      ELSE
         D = .FALSE.
      END IF
*
      SELECT = D
*
      RETURN
      END
```

9.2 Program Data

```
F08PAF Example Program Data

  4                          :Value of N

  0.35   0.45  -0.14  -0.17
  0.09   0.07  -0.54   0.35
 -0.44  -0.33  -0.03   0.17
  0.25  -0.32  -0.13   0.11 :End of matrix A
```

9.3 Program Results

```
 F08PAF Example Program Results

 Number of eigenvalues for which SELECT is true =    2

 Schur matrix T
         1       2       3       4
1   0.7995 -0.0059  0.0751 -0.0927
2   0.0000 -0.1007 -0.3937  0.3569
3   0.0000  0.0000 -0.0994  0.5128
4   0.0000  0.0000 -0.3132 -0.0994

 Matrix of Schur vectors Z
         1       2       3       4
1   0.6551  0.1210 -0.5032 -0.5504
2   0.5236  0.3286  0.7857 -0.0229
3  -0.5362  0.5974  0.0904 -0.5894
4   0.0956  0.7215 -0.3482  0.5908
```